Sistemas de EDOs. Oscilaciones acopladas. Estabilidad

■ Oscilaciones acopladas. Modos normales de vibración



Consideremos el sistema de dos bloques de masas m_1 y m_2 sujetos a muelles de constantes elásticas k_1 y k_2, como se muestra en la figura anterior. La ecuación difeencial que gobierna el movimiento de ambos bloques cuando ambos se desplazan cantidades x_1 y x_2 respecto de su posición de equilibrio es:

$$\begin{cases} m_1\,\ddot{x}_1 = -(k_2+k_1)\,x_1 + k_2\,x_2 \\ m_2\,\ddot{x}_2 = k_2\,x_1 - (k_2+k_3)\,x_2 \end{cases}$$

Pongamos un nombre a dichas ecuaciones:

```
In[1]:= ec1 = m1 x1''[t] + (k2 + k1) x1[t] - k2 x2[t];
        ec2 = m2 x2''[t] - k2 x1[t] + (k2 + k3) x2[t];
```

Podemos resolver de forma exacta dichas ecuaciones diferenciales con condiciones iniciales:

posición inicial: $x_1(0) = x_1^{(0)}$, $x_2(0) = x_2^{(0)}$

velocidad inicial: $\dot{x}_1(0) = v_1^{(0)}$, $\dot{x}_2(0) = v_2^{(0)}$

Tomemos, por ejemplo, $m_1 = m_2 = 1$ y $k_1 = k_2 = k_3 = 1$.

In[3]:=
```
m1 = 1; m2 = 1; k1 = 1; k2 = 1; k3 = 1;
osciacop = DSolve[{ec1 == 0, ec2 == 0, x1[0] == x10,
   x2[0] == x20, x1'[0] == v1, x2'[0] == v2}, {x1, x2}, t]
```

Out[4]=

$$\Big\{\Big\{x1 \to \Big(\frac{1}{12} e^{-i\,\#1 - i\sqrt{3}\,\#1}\Big(i\sqrt{3}\,e^{i\,\#1}\,v1 + 3\,i\,e^{i\sqrt{3}\,\#1}\,v1 - 3\,i\,e^{2\,i\,\#1 + i\sqrt{3}\,\#1}\,v1 - i\sqrt{3}\,e^{i\,\#1 + 2\,i\sqrt{3}\,\#1}\,v1 - i\sqrt{3}\,e^{i\,\#1}\,v2 + 3\,i\,e^{i\sqrt{3}\,\#1}\,v2 - 3\,i\,e^{2\,i\,\#1 + i\sqrt{3}\,\#1}\,v2 + i\sqrt{3}\,e^{i\,\#1 + 2\,i\sqrt{3}\,\#1}\,v2 + 3\,e^{i\,\#1}\,x10 + 3\,e^{i\sqrt{3}\,\#1}\,x10 + 3\,e^{2\,i\,\#1 + i\sqrt{3}\,\#1}\,x10 + 3\,e^{i\,\#1 + 2\,i\sqrt{3}\,\#1}\,x10 - 3\,e^{i\,\#1}\,x20 + 3\,e^{i\sqrt{3}\,\#1}\,x20 + 3\,e^{2\,i\,\#1 + i\sqrt{3}\,\#1}\,x20 - 3\,e^{i\,\#1 + 2\,i\sqrt{3}\,\#1}\,x20\Big)\ \&\Big),$$

$$x2 \to \Big(\frac{1}{12} e^{-i\,\#1 - i\sqrt{3}\,\#1}\Big(-i\sqrt{3}\,e^{i\,\#1}\,v1 + 3\,i\,e^{i\sqrt{3}\,\#1}\,v1 - 3\,i\,e^{2\,i\,\#1 + i\sqrt{3}\,\#1}\,v1 + i\sqrt{3}\,e^{i\,\#1 + 2\,i\sqrt{3}\,\#1}\,v1 + i\sqrt{3}\,e^{i\,\#1}\,v2 + 3\,i\,e^{i\sqrt{3}\,\#1}\,v2 - 3\,i\,e^{2\,i\,\#1 + i\sqrt{3}\,\#1}\,v2 - i\sqrt{3}\,e^{i\,\#1 + 2\,i\sqrt{3}\,\#1}\,v2 - 3\,e^{i\,\#1}\,x10 + 3\,e^{i\sqrt{3}\,\#1}\,x10 + 3\,e^{2\,i\,\#1 + i\sqrt{3}\,\#1}\,x10 - 3\,e^{i\,\#1 + 2\,i\sqrt{3}\,\#1}\,x10 + 3\,e^{i\,\#1}\,x20 + 3\,e^{i\sqrt{3}\,\#1}\,x20 + 3\,e^{2\,i\,\#1 + i\sqrt{3}\,\#1}\,x20 + 3\,e^{i\,\#1 + 2\,i\sqrt{3}\,\#1}\,x20\Big)\ \&\Big)\Big\}\Big\}$$

Las frecuencias naturales (al cuadrado y cambiadas de signo) de oscilación y los modos normales de vibración son los valores propios y vectores propios, respectivamente, de la matriz de coeficientes

$$\begin{pmatrix} -(k_2+k_1)/m_1 & k_2/m_1 \\ k_2/m_2 & -(k_2+k_3)/m_2 \end{pmatrix}$$

In[5]:=
```
mcoef = {{-(k2 + k1) / m1, k2 / m1}, {k2 / m2, -(k2 + k3) / m2}}
{ω1c, ω2c} = Eigenvalues[mcoef]
paso = Transpose[Eigenvectors[mcoef]]
```

Out[5]= `{{-2, 1}, {1, -2}}`

Out[6]= `{-3, -1}`

Out[7]= `{{-1, 1}, {1, 1}}`

de manera que las frecuencias propias naturales de oscilación son: $\omega_1 = \sqrt{3}$, $\omega_2 = 1$ y los modos normales y_1, y_2 se obtienen a partir de $\begin{pmatrix} x_1 \\ x_2 \end{pmatrix} = P \begin{pmatrix} y_1 \\ y_2 \end{pmatrix} \Longrightarrow \begin{pmatrix} y_1 \\ y_2 \end{pmatrix} = P^{-1}\begin{pmatrix} x_1 \\ x_2 \end{pmatrix}$, donde P es la matriz de paso. o sea:

In[8]:=
```
{y1[t], y2[t]} = Inverse[paso].{x1[t], x2[t]}
```

Out[8]= $\left\{-\frac{x1[t]}{2} + \frac{x2[t]}{2},\ \frac{x1[t]}{2} + \frac{x2[t]}{2}\right\}$

Este cambio de función incógnita nos desacopla el sistema de ecuaciones quedándo simplemente:

$$\ddot{y}_1 = -\omega_1{}^2\, y_1$$
$$\ddot{y}_2 = -\omega_2{}^2\, y_2$$

Para "estimular" el modo 1 debemos hacer $y_2 = 0$, lo cual se consigue tomando como condiciones iniciales, por ejemplo:

posición inicial: $x_1(0) = -x_2(0) = a = 1$

velocidad inicial: $\dot{x}_1(0) = 0$, $\dot{x}_2(0) = 0$

Introduzcamos estas condiciones iniciales

```
In[9]:= x10 = 1; x20 = -1; v1 = 0; v2 = 0;
```

y representemos este modo de vibración:

```
In[10]:=
<< Graphics`Legend`
Plot[{Evaluate[x1[t] /. osciacop], Evaluate[x2[t] /. osciacop]},
 {t, 0, 5}, PlotStyle -> {GrayLevel[0.], Dashing[{0.02, 0.02}]},
 PlotLegend -> {"x1[t]", "x2[t]"}]
```




```
Out[11]= - Graphics -
```

Así hemos estimulado el modo ANTISIMÉTRICO y_1 (los bloques se separan y se juntan de forma periódica) que tiene una frecuencia $\omega_1 = \sqrt{3}$ mayor que la del modo y_2 (y, por lo tanto mayor energía ¿por qué?).

Para estimular el modo y_2 debemos hacer $y_1 = 0 \Longrightarrow$

posición inicial: $x_1(0) = x_2(0) = 1$

velocidad inicial: $\dot{x}_1(0) = 0$, $\dot{x}_2(0) = 0$

Introduciendo estas condiciones iniciales

```
In[12]:= x10 = 1; x20 = 1; v1 = 0; v2 = 0;
```

y representemos este modo de vibración:

In[13]:=
```
<< Graphics`Legend`
Plot[{Evaluate[x1[t] /. osciacop], Evaluate[x2[t] /. osciacop]},
 {t, 0, 5}, PlotStyle -> {GrayLevel[0.], Dashing[{0.02, 0.02}]},
 PlotLegend -> {"x1[t]", "x2[t]"}]
```



Out[14]= `- Graphics -`

vemos que ambos bloques se mueven al unísono (modo SIMÉTRICO), resultando un movimiento de menor frecuencia y, por lo tanto, de menor energía.

Ejercicio

Resuelva el problema anterior para $k_1 = k_3 = 1$, $k_2 = 2$ y $m_1 = 2\, m_2 = 1$. Identifique las frecuencias propias y los modos normales de vibración. Elija condiciones iniciales adecuadas para estimular el modo 1 y represén-telo gráficamente. Haga lo mismo para el modo 2.

Ejercicio

Considere un bloque de 7 pisos en el que las oscilaciones transversales del terreno inducen un movimiento horizontal en cada uno de los pisos, de forma que el piso número *i* está acoplado al número *i+1* y al *i-1* mediante la ecuación

$$m_i\, \ddot{x}_i = k(x_{i+1} - x_i) - k(x_i - x_{i-1})$$

Cada piso pesa *m=16* toneladas y existe una fuerza horizontal restauradora interna entre cada piso con constante elástica *k=160* toneladas por decímetro. Calcular las 7 frecuencias naturales propias de oscilación. ¿Entraría en resonancia el edificio con un temblor de tierra de frecuencia $\omega \approx 7$ seg^{-1}? ¿Qué modo entra en resonancia con una frecuencia $\omega \approx 2$ seg^{-1}?

■ Estabilidad

Nos planteamos estudiar la estabilidad de un sistema de ecuaciones de primer orden como:

$$\begin{cases} \dot{x} = x + 3\, y \\ \dot{y} = -x \end{cases}$$

La matriz del sistema es:

In[15]:= `A = {{1, 3}, {-1, 0}};`

Y su traza y su determinante son:

```
In[16]:= Tr[A]
         Det[A]

Out[16]= 1

Out[17]= 3
```

Luego tenemos una espiral o foco inestable. Definamos una familia de soluciones:

```
In[18]:= Sol[t_, x0_, y0_] :=
           {x[t], y[t]} /. DSolve[{x'[t] == x[t] + 3 y[t], y'[t] == -x[t],
              x[0] == x0, y[0] == y0}, {x[t], y[t]}, t]
```

Tomemos unos cuantos miembros de esta familia biparamétrica de soluciones en una tabla

```
In[19]:= familia = Table[Sol[t, x0, y0], {x0, -2, 2, 2}, {y0, -2, 2, 2}];
```

y representémosla en un diagrama

```
In[20]:= curvas = ParametricPlot[Evaluate[Flatten[familia, 2]], {t, -3, 3}]
```




```
Out[20]= - Graphics -
```

Vemos claramente que tenemos un punto espiral. Para verificar que, efectivamente, es inestable, realizaremos una representación del campo de velocidades:

```
In[21]:= << Graphics`PlotField`
         velocidades = PlotVectorField[{x + 3 y, -x}, {x, -10, 10}, {y, -7, 7}];
```



que apunta claramente hacia afuera del punto espiral (inestable). Superpongamos ambos gráficos:

In[23]:= `Show[curvas, velocidades]`



Out[23]= `- Graphics -`

Ejercicio

Dado el sistema $\left\{ \begin{array}{c} \dot{x} = x \\ \dot{y} = -x + 2\,y \end{array} \right.$ determine qué tipo de punto crítico es *(x,y)=(0,0)* (nodo, vórtice, espiral o punto de silla) y diga si es estable, inestable o asintóticamente estable. Realize un gráfico con una familia de soluciones y otro con el campo de velocidades y superpóngalos (como en el ejercicio anterior).